Van Arkel Short Strangle 2 

{- Filename: Short Strangle 2 -}

{ met instelbare expiratiedag }

var
  DoW: integer;

function IsNthDoW(DoW, n, j: integer): boolean;
var
  x, Yr, Mon, Day: integer;
begin
  if (j < BarCount-1) and (DayOfWeek(DateTime[j]) < DoW)
      and (DayOfWeek(DateTime[j]) > DayOfWeek(DateTime[j + 1])) then
    x:= trunc(DateTime[j]) + DoW - DayOfWeek(DateTime[j])
    else
      x:= trunc(DateTime[j]);
  DecodeDate(x, Yr, Mon, Day);
  Result:= (DayOfWeek(x) = DoW) and (Day >= (n - 1) * 7 + 1) and (Day < n * 7 + 1);
end;

function NextNthDoW(DoW, n: integer; ADateTime: TDateTime): TDateTime;
{ Berekent de eerste n-de vrijdag op of na ADateTime }
var
  Yr, Mon, Day, DW, NextDoW, DoWCount: integer;
begin
  DecodeDate(ADateTime, Yr, Mon, Day);
  DW := DayOfWeek(ADateTime);
  if DW <= 5 then
    NextDoW:= Day + (DoW - DW)
  else
    NextDoW:= Day + DoW;

  DoWCount:= (NextDoW + DoW) div 7;
  if DoWCount > n then
  begin
    if Mon < 12 then
      Mon:= Mon + 1
    else
    begin
      Mon:= 1;
      Yr:= Yr + 1;
    end;
    Result:= NextNthDoW(DoW, n, EncodeDate(Yr, Mon, 1));
  end else
    Result:= trunc(ADateTime) + (3 - DoWCount) * 7 + NextDoW - Day;
end;

function TradingDaysToExpiry(n: integer): TSeries;
var
  i, count, weeks, days: integer;
begin
  Result:= CreateSeries(BarCount);
  for i:=0 to BarCount - 1 do
  begin
    count:= trunc(NextNthDoW(Dow, n, DateTime[i])) - trunc(DateTime[i]);
    weeks:= count div 7;
    days:= count mod 7;
    Result[i]:= weeks * 5 + days;
  end;
end;

var 
  i, Length, nLong, nShort, nLongDaysToExp, nShortDaysToExp: integer;
  iOldLongFriBeforeExp, iOldShortFriBeforeExp, iLongFriBeforeExp, iShortFriBeforeExp: integer;
  nLongTrades, nShortTrades, nLongStop, nShortStop, Treshold1, Treshold2: integer;
  sVolatility, sStopLong, sStopShort, sLongDaysToExpiry, sShortDaysToExpiry: TSeries;
  OldLongDate, OldShortDate, NewLongDate, NewShortDate: TDateTime;
  OldUpVal, OldDnVal, UpValue, DnValue, FactLong, FactShort, VltTreshold, VltMin, ShortStopPerc: real;
  Rnd, Rnd1, Rnd2, Rnd3, LongStop, ShortStop, LongStopDelta, ShortStopDelta, FactStopDelta: real;
  UpColor, DnColor: TColor;
  xStats, StopTrades: Boolean;

begin
  DoW:= 1 + CreateParameterInteger('Expiration Day [Monday = 1]', 1, 5, 5, false);
  nLong:= 3 - CreateParameterInteger('Bullish trade wks before Exp.', 0, 2, 0, false);
  nShort:= 3 - CreateParameterInteger('Bearish trade wks before Exp.', 0, 2, 0, false);
  Length:= CreateParameterInteger('Volatility Length', 1, 999, 22, true);
  VltTreshold:= CreateParameterReal('Volatility treshold', 0, 100, 100, true);
  FactLong:= CreateParameterReal('Volatility factor bullish', 0, 9.99, 0.432, true);
  FactShort:= CreateParameterReal('Volatility factor bearish', 0, 9.99, 0.432, true);
  Treshold2:= CreateParameterInteger('If price >', Treshold1, 10000, 200, false);
  Rnd3:= CreateParameterReal('Round to pt.', Rnd2, 100, 10, false);
  Treshold1:= CreateParameterInteger('If price >', 1, 10000, 100, false);
  Rnd2:= CreateParameterReal('Round to pt.', Rnd1, 100, 5, false);
  Rnd1:= CreateParameterReal('Else round to pt.', 0, 1000, 2.5, false);
  ShortStopPerc:= CreateParameterReal('Stoploss bearish as % of strike', 0, 100, 0.5, true)/100;
  FactStopDelta:= CreateParameterReal('Stoploss delta % of the spread', 0, 50, 0, true)/100;
  VltMin:= CreateParameterReal('Volatility minimum', 0, 100, 0, true);
  xStats:= CreateParameterBoolean('Show statistics', true, false);
    
  with Indicator do 
  begin
    ShortName:= 'VA SS';
    NewBand:= false;
    Indicator.ScaleRange:= srCommon;
    SignalView:= svShowInMain;
    RequiredBars:= Length;
  end;
  
   sVolatility:= Volatility(Close, Length, BarInterval);
   sLongDaysToExpiry:= TradingDaysToExpiry(nLong);
   sShortDaysToExpiry:= TradingDaysToExpiry(nShort);
   sStopLong:= CreateSeries(BarCount);
   sStopShort:= CreateSeries(BarCount);

  if BarCount > 0  then
  begin
    NewLongDate:= DateTime[0];
    NewShortDate:= DateTime[0];
    UpValue:= High[0];
    DnValue:= Low[0];
    iOldLongFriBeforeExp:= 0;
    iOldShortFriBeforeExp:= 0;
    iLongFriBeforeExp:= 0;
    iShortFriBeforeExp:= 0;
    nLongTrades:= 0;
    nShortTrades:= 0;
    nLongStop:= 0;
    nShortStop:= 0;
    StopTrades:= false;

    for i:= firstvalidindex(sVolatility) to BarCount - 1 do
    begin
      if IsNthDoW(DoW, 3, i) then //Epiry day
        Signals[i]:= sgMark;
      if IsNthDoW(DoW, Trunc(Min(nLong, nShort)), i) then
        if sVolatility[i] > VltTreshold then
          StopTrades:= true else StopTrades:= false;
      if IsNthDoW(DoW, nLong, i - 1) then
      begin
        nLongDaysToExp:= Trunc(sLongDaysToExpiry[i]);
        LongStop:= UpValue;
        if StopTrades then
          LongStopDelta:= 0 else
            LongStopDelta:= FactStopDelta * (FactLong + FactShort) * sVolatility[i] / 100 * Close[i];
        if StopTrades then
          UpColor:= clWhite else UpColor:= clAqua;
        if sVolatility[i] < VltMin then UpColor := clYellow;
      end;
      if IsNthDoW(DoW, nShort, i - 1) then
      begin
        nShortDaysToExp:= Trunc(sShortDaysToExpiry[i]);
        ShortStop:= DnValue * (1 + ShortStopPerc);
        if StopTrades then
          ShortStopDelta:= 0 else
            ShortStopDelta:= FactStopDelta * (FactLong + FactShort) * sVolatility[i] / 100 * Close[i];
        if StopTrades then
          DnColor:= clWhite else DnColor:= clAqua;
        if sVolatility[i] < VltMin then DnColor := clYellow;
      end;
      if IsValid(sLongDaysToExpiry[i]) and IsValid(sShortDaysToExpiry[i])
        and (nLongDaysToExp <> 0) and (nShortDaysToExp <> 0) then
      begin
        sStopLong[i]:= LongStop - LongStopDelta * sLongDaysToExpiry[i] / nLongDaysToExp;
        sStopShort[i]:= ShortStop + ShortStopDelta * sShortDaysToExpiry[i] / nShortDaysToExp;
//        if (Close[i] > sStopLong[i]) and (UpColor = clAqua) then UpColor := clRed;
//        if (Close[i] < sStopShort[i]) and (DnColor = clAqua) then DnColor := clRed;
        if (High[i] > sStopLong[i]) and (UpColor = clAqua) then UpColor := clRed;
        if (Low[i] < sStopShort[i]) and (DnColor = clAqua) then DnColor := clRed;
      end;
      if Close[i] < Treshold1 then Rnd:= Rnd1 else
        if Close[i] < Treshold2 then Rnd:= Rnd2 else Rnd:= Rnd3;
      if IsNthDoW(DoW, nLong, i) then
      begin
        SetParentBarColor(i, clBlue);
        OldUpVal:= UpValue;
        OldLongDate:= NewLongDate;
        iOldLongFriBeforeExp:= iLongFriBeforeExp;
        iLongFriBeforeExp:= i;
        NewLongDate:= datetime[i];
        if not (UpColor in [clWhite, clYellow]) then nLongTrades:= nLongTrades + 1;
        if UpColor = clRed then nLongStop := nLongStop + 1;
        UpValue:= Rnd * Round(Close[i] * 0.01 * (100 + FactLong * sVolatility[i]) / Rnd);
        with CreateTrendline(OldLongDate, OldUpVal, NewLongDate, OldUpVal) do
        begin
          ExtendLeft  := false;
          ExtendRight := false;
          Width       := 2;
          Color       := UpColor;
        end;
        with CreateTrendline(NewLongDate, OldUpVal, NewLongDate, UpValue) do
        begin
          ExtendLeft  := false;
          ExtendRight := false;
          Width       := 1;
          Color       := clAqua;
        end;
      end;
      if IsNthDoW(DoW, nShort, i) then
      begin
        SetParentBarColor(i, clRed);
        OldDnVal:= DnValue;
        OldShortDate:= NewShortDate;
        iOldLongFriBeforeExp:= iShortFriBeforeExp;
        iShortFriBeforeExp:= i;
        NewShortDate:= datetime[i];
        if not (DnColor in [clWhite, clYellow]) then nShortTrades:= nShortTrades + 1;
        if DnColor = clRed then nShortStop := nShortStop + 1;
        DnValue:= Rnd * Round(Close[i] * 0.01 * (100 - FactShort * sVolatility[i]) / Rnd);
        with CreateTrendline(OldShortDate, OldDnVal, NewShortDate, OldDnVal) do
        begin
          ExtendLeft  := false;
          ExtendRight := false;
          Width       := 2;
          Color       := DnColor;
        end;
        with CreateTrendline(NewShortDate, OldDnVal, NewShortDate, DnValue) do
        begin
          ExtendLeft  := false;
          ExtendRight := false;
          Width       := 1;
          Color       := clAqua;
        end;
      end;

      if i= barcount - 1 then
      begin
        if not (UpColor in [clWhite, clYellow]) then nLongTrades:= nLongTrades + 1;
        if not (DnColor in [clWhite, clYellow]) then nShortTrades:= nShortTrades + 1;
        if UpColor = clRed then nLongStop:= nLongStop + 1;
        if DnColor = clRed then nShortStop:= nShortStop + 1;
        with CreateTrendline(NewShortDate, DnValue,datetime[i], DnValue) do
        begin
          ExtendLeft  := false;
          ExtendRight := true;
          Width       := 2;
          Color       := DnColor;
        end;
        with CreateTrendline(NewLongDate, UpValue, datetime[i], UpValue) do
        begin
          ExtendLeft  := false;
          ExtendRight := true;
          Width       := 2;
          Color       := UpColor;
        end;
      end;
    end;
  end;
  with Createline(sStopLong) do
  begin
    Width:= 1;
    Color:= clRed;
    LineType:= ltDot;
  end;
  with Createline(sStopShort) do
  begin
    Width:= 1;
    Color:= clRed;
    LineType:= ltDot;
  end;
  if xStats and (nLongTrades <> 0) and (nShortTrades <> 0) then
    with CreateText(0, 0,
      'Stats. Bullish:' + #10 +
      'Trades:  ' + IntToStr(nLongTrades) + #13 +
      'Stops:    ' + IntToStr(nLongStop) + #13 +
      'Hitrate:   ' + IntToStr(100 - Round(100 * nLongStop / nLongTrades)) + '%' + #13 +
      'Bearish:' + #13 +
      'Trades:  ' + IntToStr(nShortTrades) + #13 +
      'Stops:    ' + IntToStr(nShortStop) + #13 +
      'Hitrate:   ' + IntToStr(100 - Round(100 * nShortStop / nShortTrades)) + '%') do
    begin
      X1Pct:= 92.5;
      Y1Pct:= 100;
      Color:= clYellow;
    end;
END.

